Lattice Boltzmann for flow and transport phenomena

1. Introduction to the lattice Boltzmann method

Li Chen

2018/05

Key Laboratory of Thermo-Fluid Science and Engineering of MOE

Xi’an Jiaotong University, Xi’an China

Mail: lichennht08@mail.xjtu.edu.cn

http://gr.xjtu.edu.cn/web/lichennht08

Content

1.1 Background



Micro, meso and macroscopic models and equations

1.1 Background

Microscopic models require huge computational resources.

(1mol, $6.02*10^{23}$)

Macroscopic models cannot provide underlying details.

Chang-Lin Tien (Microscale Thermophysical Engineering: 1997, 1: 71~84)

(1935-2002, 7th president of UC Berkeley, 1990-1997)

Many physical phenomena and engineering problems may have their origins at molecular scales, although they need to interface with the macroscopic or "human" scales. The difficulty arises in bridging the results of these models across the span of length and time scales. The lattice Boltzmann method attempts to bridge this gap.（联系微观和宏观的桥梁）

Mesoscopic scale

What is mesoscopic scale （Mesoscale) ?

The scale between microscale and macroscale.

Proposed by VanKampen 1981.

The size of this scale is comparable to the macroscale, yet lots of transport phenomena which we thought only take place at the microscale also can be observed at this scale.

Serve as a bridge between the gas of microscale and macroscale!

1.1 Background

The lattice Boltzmann method has been adopted for flow and transport phenomena in a wide range of scientific and engineering problem, especially for porous media flow and multiphase flow

- **Porous flow**











Li Chen, et al, Scientific reports,2015

1.1 Background

- **Multiphase flow**









Li Chen, et al, IJHMT, 2014

1.1 Background

● Reactive transport









Li Chen, et al, IJHE, 2012

Li Chen, et al, WRR, 2014; Langmuir, 2012

1.1 Background











Li Chen, et al, Electrochemica Acta, 2015

Adsorption of SVOC in PM 2.5











2016, YT Mu, et al., BAE

● Heat transfer

二氧化硅气凝胶有效导热系数





2015 WZ Fang, et al., IJHMT

在航空航天、保温节能等领域具有广泛的应用前景!

● Heat transfer (Boiling)



Density (left) and temperature (right) distribution

1.1 Background

- **Particle flow**





K. Connington, Q. Kang, H. Viswanathan, A. Abdel-Fattah, S. Chen, Peristaltic particle transport using the lattice Boltzmann method, Physics of Fluids, 21(5) (2009) 053301

1.1 Background

- **Crystal growth**



(a) (b)

(c) (d)



Dendrite



Q. Kang, D. Zhang, P.C. Lichtner, I.N. Tsimpanogiannis, Lattice Boltzmann model for crystal growth from supersaturated solution, Geophysical research letters, 31 (2004) L21604.


Prob 100
Prob 75
Prob 50
Prob 25
Prob 12.5
Prob 0
Df 1.55
220 µm


Da 4 Df 1.74
Sat 1.15
1.5 µm
Da 96 Df 1.28
35 µm
Da 16 Df 1.65
6 µm
Da 150 Df 1.23
55 µm
Da 48 Df 1.43
17 µm
Da 302 Df 1.21
110 µm

Snow

● Reactive transport & Crystal growth



（a）针状沉淀 （b）树状沉淀 （c）过渡沉淀图案

Diverse precipitate patterns predicted







Li Chen, et al, Langmuir, 2012

1.1 Background

Inception phase: 1988-1992

The predecessor of LBM is LGA (lattice gas automata)

1988: McNamara and Zanetti proposed to use real numbers rather than Boolean value “0” and “1” (PRL, 1988)

- **Avoid statistic noise**
- **Becomes complicated with multiple particles collision at one site.**

1989: Higuere and Jinenes developed the linear collision term. (Europhys. Lett, 1989)

1991-1992: several groups simultaneously proposed the BGK collision term or SRT (single relaxation time) collision term.

The purpose of collision to approach equilibrium state.

(S. Chen et al. PRL, 1991; Y. Qian et al. Europhys. Lett 1992)

1.1 Background

Development Phase: 1988-1998

Several groups proved that LBM can be rigorously derived from Boltzmann equation

(T. Abe. JCP 1997; X. He and L-S Luo 1997 PRE; X. Shan and X. He. PRL, 1998)

Heat Transfer:

- **Double distribution** (X. Shan 1997 PRE, X. He, S. Chen, G. D. Doolen. JCP 1998)
- **Multi-speed model**

Multiphase and multicomponent flow:

- **Pseudopotential model (X. Shan, H. Chen, PRE, 1993)**
- **Color model (A. K. Gunstensen, et al., PRA, 1991)**
- **Free energy model (Swift et al. PRL, 1995)**

1.1 Background

Particle flow:

(A. J. C. Ladd, PRL, 1993, JFM 1994, JSP, 1995)

Reaction:

S. P. Dawson et al. J. Chem. Phys. 1993

S. Succi, G. Bella and F. Papetti, J. Sci. Comput. 1997

Other complex flow:

Non-newtonian fluid flow, magnetic fluid, blood flow, polymeric flow……

Boundary condition

Porous media flow （渗流）

1.1 Background

Rapid Development Phase: 1999- present

It has been paid great attention both on theory development and Engineering application. High *Re* flow, multiphase flow with large density and viscosity ratio, turbulent flow, combustion, three-phase flow, phase change heat transfer (boiling, condensation, melting, solidification), multicomponent reactive transport, slip flow, electro osmotic flow, MRT LB model…..

Commercial software is developed, such as Power Flow

Now it has been developed as an powerful alternative tool for flow and transport process, especially for that in complex structures and multiphase flow.

Content

1.2 Boltzmann equation

Velocity distribution function

$$f\,(x,\xi,t)$$

Number of molecules is $f dx d\xi$ in control volume dx with velocity in the range of (ξ, ξ +dξ) at time t.

Thus, the total molecules in the control volume dx is

$$\int f d\xi = n$$

Correspondingly, the total mass, momentum and energy are

$$\int m f d\xi = \rho$$

$$\int m\xi f d\xi = \rho u$$

$$\int m\frac{\xi^2}{2} f d\xi = \rho E = \rho e + \frac{1}{2}\rho u^2$$

1.2 Boltzmann equation

For a certain molecular, if during the time *dt*, it does not collision with other molecules, its position and velocity will change as follows

$$x+\xi dt, \quad \xi + adt$$

Therefore, at time *t+dt*, those molecules in (*x*, *ξ*) at time *t* will move to (*x+dx*, *ξ +dξ*)

$$f(\mathbf{x}+d\mathbf{x}, \boldsymbol{\xi}+adt, t+dt)\,d\mathbf{x}\,d\boldsymbol{\xi} = f(\mathbf{x}, \boldsymbol{\xi}, t)\,d\mathbf{x}\,d\boldsymbol{\xi}$$



1.2 Boltzmann equation

$$f(\boldsymbol{x}+d\boldsymbol{x},\xi+adt,t+dt)d\boldsymbol{x}d\xi = f(\boldsymbol{x},\xi,t)d\boldsymbol{x}d\xi$$

Taylor expansion

$$\frac{\partial f}{\partial t}+\xi\cdot\frac{\partial f}{\partial \boldsymbol{x}}+a\cdot\frac{\partial f}{\partial \xi}=0$$

The above equation is the Boltzmann equation without collision. It describes the conservation of velocity distribution function.

However, collision between molecules also change the velocity of molecules, thus a collision term should be added

1.2 Boltzmann equation

Collision between molecules also change the velocity of molecules, thus a collision term should be added

$$\left(\frac{\partial f}{\partial t}\right)_{\text{collision}}$$

$$\frac{\partial f}{\partial t}+\xi\frac{\partial f}{\partial \boldsymbol{x}}+a\cdot\frac{\partial f}{\partial \xi}=\left(\frac{\partial f}{\partial t}\right)_{\text{collision}}=\Omega(f)$$

Collision between molecules are complex. Even if we assume two-molecule collision, velocities are uncorrelated pre-collision, at during the short time of collision, external force does not play a role

$$\left(\frac{\partial f}{\partial t}\right)_{\text{collision}}=\int\int(f' f_1'-ff_1)d_D^{\,2}\,|g|\cos\theta d\Omega d\xi_1$$

1.2 Boltzmann equation

Boltzmann equation

$$\frac{\partial f}{\partial t}+\xi\frac{\partial f}{\partial \boldsymbol{x}}+a\cdot\frac{\partial f}{\partial \xi}=\int\int(f'f_1'-ff_1)d_D^{\;2}|g|\cos\theta d\Omega d\xi_1$$

Devised by Ludwig Boltzmann in 1872.

The collision term is a integral-differential term. Thus, it is really hard to solve Boltzmann equation due to this complex term.

Simplify the collision term: Boltzmann *H* theorem

$$H=\overline{\ln f}=\frac{\int f\ln f\,\mathrm{d}\xi}{\int f\,\mathrm{d}\xi}=\frac{1}{n}\int f\ln f\,\mathrm{d}\xi$$

1.2 Boltzmann equation

Boltzmann *H* theorem

$$\partial H / \partial t \le 0$$

Indicating that *H* of a system decreases with the time. The final state means the equilibrium system. The corresponding *f* is equilibrium distribution function. Maxwell distribution is a solution



$$f^{eq} = \frac{1}{(2\pi RT)^{3/2}} \exp\left(-\frac{(\xi - u)^2}{2RT}\right)$$

In this mechanical model of a gas, the motion of the molecules appears very disorderly. Boltzmann showed that, assuming each collision configuration in a gas is truly random and independent, the gas converges to the Maxwell speed distribution even if it did not start out that way.

1.2 Boltzmann equation

BGK collision term: In 1954, **B**hatnagar, **G**ross and **K**rook proposed the BGK collision model（SRT)

$$-\frac{1}{\tau}(f-f^{\mathrm{eq}})$$

where τ is the relaxation time, f^{eq} is the equilibrium distribution function.

1) Approximate that the effect of collision is to force the non-equilibrium distribution back to Maxwell equilibrium distribution

2) The collision term should maintain the conservation of mass, momentum and energy.

$$\frac{\partial f}{\partial t}+\xi\cdot\frac{\partial f}{\partial \mathbf{x}}+a\cdot\frac{\partial f}{\partial \xi}=-\frac{1}{\tau}(f-f^{\mathrm{eq}})$$

1.2 Boltzmann equation

Velocity distribution function

$$f(x,\xi,t)$$

Boltzmann Equation

$$\frac{\partial f}{\partial t}+\xi\cdot\frac{\partial f}{\partial \mathbf{x}}+a\cdot\frac{\partial f}{\partial \xi}=-\frac{1}{\tau}(f-f^{\mathrm{eq}})$$

BGK (SRT) collision term

Equilibrium distribution function

$$f^{\mathrm{eq}}=\frac{1}{(2\pi RT)^{3/2}}\exp\left(-\frac{(\xi-\mathrm{u})^2}{2RT}\right)$$

Content

1.3 Lattice Boltzmann method

Lattice Boltzmann method

From lattice gas automata
1973-1988
HPP
FHP

Discretization of
Boltzmann equation
1997~1998

1.3.1 Lattice Gas Automata

Hardy, de Pazzls, Pomeau, 1973 HPP



$C_1=(1,0)$
$C_2=(0,1)$
$C_3=(-1,0)$
$C_4=(0,-1)$

Head on collision



1. **Pauli incompatible principle**

   $n(x,t)=n_1 n_2 n_3 n_4$

$$c_i(n) = n_{i\oplus 1} n_{i\oplus 3}(1-n_i)(1-n_{i\oplus 2}) - (1-n_{i\oplus 1})(1-n_{i\oplus 3}) n_i n_{i\oplus 2}$$

2. **Collision and streaming**

1.3.1 Lattice Gas Automata

Frisch, Hasslacher, Pomeau, 1986 FHP



NS equation can be recovered.

Because of Boolean number, statistic noise is huge.

The collision operator is complex.

1988: McNamara and Zanetti proposed to use real numbers rather than Boolean value "0" and "1" (PRL, 1999)

- **Avoid statistic noise**
- **Becomes complicated with multiple particles collision at one site.**

1989: Higuere and Jinenes developed the linear collision term. (Europhys. Lett)

1991-1992: several groups simultaneously proposed the BGK collision term or SRT collision term.

The purpose of collision to approach equilibrium state. (S. Chen et al. PRL, 1991; Y. Qian et al. Europhys. Lett 1992)

1.3.2 Discretization of Boltzmann equation

Discretization of velocity, space and time of Boltzmann equation

Velocity is continuous, however it is impossible to consider all the PDF in all velocity directions.

The velocity is discretized. The most famous model is DnQm lattice model. n denotes dimension, and m is the number of velocity directions. (Q: Yuehong Qian, 钱跃竑老师).



Infinity to just 9

$$f \to f_i$$



1.3.2 Discretization of Boltzmann equation

$$\mathbf{c}_0 = \mathbf{0} \quad ,$$
$$\mathbf{c}_i = \left(\cos\frac{i-1}{2}\pi, \sin\frac{i-1}{2}\pi\right)\frac{\Delta x}{\Delta t} \quad , \quad i = 1-4 \quad ,$$
$$\mathbf{c}_i = \sqrt{2}\left(\cos\frac{2i-9}{4}\pi, \sin\frac{2i-9}{4}\pi\right)\frac{\Delta x}{\Delta t} \quad , \quad i = 5-8$$



$$f^{\mathrm{eq}} = \frac{\rho}{(2\pi RT)^{3/2}}\exp\left(-\frac{(\xi - \mathrm{u})^2}{2RT}\right)$$

$$e^x = \sum_{n=0}^{\infty}\frac{x^n}{n!} = 1 + x + \frac{x^2}{2!} + \frac{x^3}{3!} + \cdots \quad \text{for all } x$$

Expanded around u:

$$f^{\mathrm{eq}} = \frac{\rho}{(2\pi RT)^{3/2}}\exp\left(-\frac{\xi^2}{2RT}\right)\left(1 + \frac{\xi u}{2RT} + \frac{(\xi u)^2}{2(RT)^2} - \frac{(u)^2}{2(RT)^2}\right)$$

1.3.2 Discretization of Boltzmann equation

Equilibrium distribution function

$$f_i^{eq} = \rho w_i \left[1 + \frac{\mathbf{u}\cdot\mathbf{c}_i}{c_s^2} + \frac{1}{2}\left(\frac{\mathbf{u}\cdot\mathbf{c}_i}{c_s^2}\right)^2 - \frac{\mathbf{u}\cdot\mathbf{u}}{2c_s^2} \right]$$

$$\sum f_i^{eq} = \sum f^{eq} = \rho$$

$$\sum f_i^{eq}\mathbf{c}_i = \sum f^{eq}\boldsymbol{\xi} = \rho\mathbf{u}$$

$$\sum f_i^{eq}\mathbf{c}_i\mathbf{c}_i = \sum f^{eq}\boldsymbol{\xi}\boldsymbol{\xi} = \rho\mathbf{u}\mathbf{u} + p\mathbf{I}$$

$$\sum_{i=1}^{4}\mathbf{c}_{i\alpha} = 0 \qquad \sum_{i=5}^{8}\mathbf{c}_{i\alpha} = 0 \qquad \sum_{i=1}^{4}\mathbf{c}_{i\alpha}\mathbf{c}_{i\beta}\mathbf{c}_{i\gamma} = 0 \qquad \sum_{i=5}^{8}\mathbf{c}_{i\alpha}\mathbf{c}_{i\beta}\mathbf{c}_{i\gamma} = 0$$

$$\sum_{i=1}^{4}\mathbf{c}_{i\alpha}\mathbf{c}_{i\beta} = 2\delta_{\alpha\beta} \qquad \sum_{i=5}^{8}\mathbf{c}_{i\alpha}\mathbf{c}_{i\beta} = 4\delta_{\alpha\beta} \qquad \sum_{i=1}^{4}\mathbf{c}_{i\alpha}\mathbf{c}_{i\beta}\mathbf{c}_{i\gamma}\mathbf{c}_{i\chi} = 2\delta_{\alpha\beta\gamma\chi} \qquad \sum_{i=5}^{8}\mathbf{c}_{i\alpha}\mathbf{c}_{i\beta}\mathbf{c}_{i\gamma}\mathbf{c}_{i\chi} = 4\Delta_{\alpha\beta\gamma\chi} - 8\delta_{\alpha\beta\gamma\chi}$$

$$\Delta_{\alpha\beta\gamma\chi} = \delta_{\alpha\beta}\delta_{\gamma\chi} + \delta_{\alpha\gamma}\delta_{\beta\chi} + \delta_{\alpha\chi}\delta_{\beta\gamma}$$

1.3.2 Discretization of Boltzmann equation

D2Q9



$$w_0 = 4/9$$
$$w_{1-4} = 1/9$$
$$w_{5-8} = 1/36$$

D3Q19



$$w_0 = 1/3$$
$$w_{1-6} = 1/18$$
$$w_{7-18} = 1/36$$

1.3.2 Discretization of Boltzmann equation

$$\frac{\partial f}{\partial t}+\xi\cdot\frac{\partial f}{\partial \mathbf{x}}+a\cdot\frac{\partial f}{\partial \xi}=-\frac{1}{\tau}(f-f^{\mathrm{eq}})$$

$$\frac{\partial f_i}{\partial t}+\xi\cdot\frac{\partial f_i}{\partial \mathbf{x}}+a\cdot\frac{\partial f_i}{\partial \xi}=-\frac{1}{\tau}(f_i-f_i^{\mathrm{eq}})$$

Evolution equation for the LBM

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t)-f_i(\mathbf{x},t)=-\frac{1}{\tau}(f_i(\mathbf{x},t)-f_i^{\mathrm{eq}}(\mathbf{x},t))$$

3 key components

- **Evolution equation**
- **Lattice model**
- **Equilibrium distribution function** f^{eq}

1.3.3 From LBM to NS equation

Fundamentally, the NS equation can be derived from the Boltzmann equation.

Champan-Ensokog expansion: C-E expansion denotes such derivation from Boltzmann equation to NS equation as well as transport coefficient from Boltzmann equation (Chapman and Enskog between 1910 and 1920)

ε Small expansion parameter

$$f_i = f_i^{(0)} + \varepsilon f_i^{(1)} + \varepsilon^2 f_i^{(2)}$$

$$\partial_{x_\alpha} = \varepsilon \partial_{x_\alpha}^{(1)}$$

$$\partial_t = \varepsilon \partial_t^{(1)} + \varepsilon^2 \partial_t^{(2)}$$

1.3.3 From LBM to NS equation

Taylor expansion

$$f_i(\boldsymbol{x}+\boldsymbol{c}_i\Delta t, t+\Delta t) = f_i(\boldsymbol{x}, t) + \Delta t D_{i\alpha} f_i(\boldsymbol{x}, t) + \frac{(\Delta t)^2}{2} D_{i\alpha}^2 f_i(\boldsymbol{x}, t) + O\left[(\Delta t)^3\right]$$

$$D_{i\alpha} = \partial_t + c_i \partial_{x_\alpha}$$

$$\varepsilon D_{i\alpha}^{(1)} f^{(0)} + \varepsilon^2 \left[D_{i\alpha}^{(1)} f_i^{(1)} + \partial_t^{(2)} f_i^{(0)}\right] + \varepsilon^2 \frac{\Delta t}{2}\left[D_{i\alpha}^{(1)}\right]^2 f_i^{(0)} = -\frac{1}{\Delta t \tau_f}\left(f_i^{(0)} + \varepsilon f_i^{(1)} + \varepsilon^2 f_i^{(2)} - f_i^{(eq)}\right) + O\left[(\Delta t)^3\right]$$

1.3.3 From LBM to NS equation

$$\varepsilon^0 : f_i^{(0)} = f_i^{(eq)}$$

$$\varepsilon^1 : f_i^{(1)} = -\Delta t \tau_f D_{i\alpha}^{(1)} f_i^{(0)} + O\left[(\Delta t)^2\right]$$

$$\varepsilon^2 : f_i^{(2)} = -\Delta t \tau_f \left[D_{i\alpha}^{(1)} f_i^{(1)} + \partial_t^{(2)} f_i^{(0)}\right] - \tau_f \frac{(\Delta t)^2}{2}\left[D_{i\alpha}^{(1)}\right]^2 f_i^{(0)} + O\left[(\Delta t)^3\right]$$

$$\varepsilon^1 : \partial_t^{(1)} \rho + \partial_{x_\alpha}^{(1)}(\rho u_\alpha) + O[(\Delta t)^2] = 0$$

$$\partial_t^{(1)}(\rho u_\alpha) + \partial_{x_\beta}^{(1)}(\rho u_\alpha u_\beta + p\delta_{\alpha\beta}) + O[(\Delta t)^2] = 0$$

$$\varepsilon^2 : \partial_t^{(2)} \rho + O\left[(\Delta t)^3\right] = 0$$

$$\partial_t^{(2)}(\rho u_\alpha) - \nu \partial_{x_\beta}^{(1)}\left\{\rho\left[\partial_{x_\alpha}^{(1)} u_\beta + \partial_{x_\beta}^{(1)} u_\alpha\right]\right\} + O\left[(\Delta t)^3\right] = 0$$

1.3.3 From LBM to NS equation

$$f_i^{(0)} = f_i^{(\mathrm{eq})}$$

$$f_i^{(1)} = -\tau_f \Delta t \left[U_{i\alpha} f_i^{(0)} \frac{1}{\rho} \partial_{x_\alpha}^{(1)} \rho + U_{i\alpha} U_{i\beta} f_i^{(0)} \frac{1}{c_s^2} \partial_{x_\alpha}^{(1)} u_\beta - U_{i\alpha} f_i^{(0)} \frac{1}{\rho c_s^2} \partial_{x_\alpha}^{(1)} p \right]$$
$$= -\tau_f \Delta t U_{i\alpha} U_{i\beta} f_i^{(0)} c_s^{-2} \partial_{x_\alpha}^{(1)} u_\beta$$

$$f_i^{(2)} = -\Delta t \tau_f \nu U_{i\beta} f_i^{(0)} c_s^{-2} \left[\frac{1}{\rho} \partial_{x_\alpha}^{(1)} \rho \left(\partial_{x_\alpha}^{(1)} u_\beta + \partial_{x_\beta}^{(1)} u_\alpha \right) + \left(\partial_{x_\alpha}^{(1)} \right)^2 u_\beta \right]$$
$$= -\Delta t \tau_f \nu U_{i\beta} f_i^{(0)} c_s^{-2} \left[\frac{1}{\rho} S_{\alpha\beta}^{(1)} \partial_{x_\alpha}^{(1)} \rho + \left(\partial_{x_\alpha}^{(1)} \right)^2 u_\beta \right]$$

$$S_{\alpha\beta} = \partial_{x_\beta} u_\alpha + \partial_{x_\alpha} u_\beta$$

1.3.3 From LBM to NS equation

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t) - f_i(\mathbf{x},t) = -\frac{1}{\tau}\left(f_i(\mathbf{x},t) - f_i^{\mathrm{eq}}(\mathbf{x},t)\right)$$

$$\upsilon = \frac{1}{3}\left(\tau - \frac{1}{2}\right)\frac{\Delta x^2}{\Delta t}$$

$$p = \rho {c_s}^2$$

$$\rho = \sum_{i=0} f_i, \qquad \rho\mathbf{u} = \sum_{i=0} f_i \mathbf{e}_i .$$

$$\frac{\partial \rho}{\partial t} + \nabla\cdot(\rho\mathbf{u}) = 0$$

$$\frac{\partial \rho\mathbf{u}}{\partial t} + \nabla\cdot(\rho\mathbf{u}\mathbf{u}) = -\nabla p + \nabla\cdot\left[\rho\upsilon(\nabla\mathbf{u} + \nabla\mathbf{u}^{\mathrm{T}}) - \frac{v}{{cs}^2}\nabla\cdot(\rho\mathbf{u})\right]$$

1.3 Lattice Boltzmann method

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t) - f_i(\mathbf{x},t) = -\frac{1}{\tau}\left(f_i(\mathbf{x},t) - f_i^{\mathrm{eq}}(\mathbf{x},t)\right)$$



Collision

$$f_i^{'}(\mathbf{x},t) = -\frac{1}{\tau}\left(f_i(\mathbf{x},t) - f_i^{\mathrm{eq}}(\mathbf{x},t)\right)$$

Streaming

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t) = f_i^{'}(\mathbf{x},t)$$

Macroscopic variables calculation

$$\rho = \sum_{i=0} f_i, \qquad \rho\mathbf{u} = \sum_{i=0} f_i\mathbf{e}_i.$$

1.3 Lattice Boltzmann method

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t) - f_i(\mathbf{x},t) = -\frac{1}{\tau}\left(f_i(\mathbf{x},t) - f_i^{\mathrm{eq}}(\mathbf{x},t)\right)$$

$$\mathbf{c}_0 = \mathbf{0} \quad ,$$

$$\mathbf{c}_i = \left(\cos\frac{i-1}{2}\pi, \sin\frac{i-1}{2}\pi\right)\frac{\Delta x}{\Delta t} \quad , \quad i = 1-4 \quad ,$$

$$\mathbf{c}_i = \sqrt{2}\left(\cos\frac{2i-9}{4}\pi, \sin\frac{2i-9}{4}\pi\right)\frac{\Delta x}{\Delta t} \quad , \quad i = 5-8$$



$$f_i^{\mathrm{eq}} = \rho w_i\left[1 + \frac{\mathbf{u}\cdot\mathbf{c}_i}{c_s^2} + \frac{1}{2}\left(\frac{\mathbf{u}\cdot\mathbf{c}_i}{c_s^2}\right)^2 - \frac{\mathbf{u}\cdot\mathbf{u}}{2c_s^2}\right]$$

$$\rho = \sum_{i=0} f_i , \qquad \rho\mathbf{u} = \sum_{i=0} f_i\mathbf{e}_i . \qquad \upsilon = \frac{1}{3}\left(\tau - \frac{1}{2}\right)\frac{\Delta x^2}{\Delta t}$$

Content

1.4 Boundary condition of the LBM

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t) - f_i(\mathbf{x},t) = -\frac{1}{\tau}\left(f_i(\mathbf{x},t) - f_i^{\mathrm{eq}}(\mathbf{x},t)\right)$$

Collision

$$f_i^{'}(\mathbf{x},t) = -\frac{1}{\tau}\left(f_i(\mathbf{x},t) - f_i^{\mathrm{eq}}(\mathbf{x},t)\right)$$

Streaming

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t) = f_i^{'}(\mathbf{x},t)$$

Macroscopic variables calculation

$$\rho = \sum_{i=0} f_i , \qquad \rho\mathbf{u} = \sum_{i=0} f_i \mathbf{e}_i .$$



1.4 Boundary condition of the LBM

Unlike traditional CFD methods such as FVM, FEM, and FDM, in the LBM the basic variable is particle distribution function (PDF)

Boundary condition is to give values to these PDF whose values are unknown after streaming step.

Since most of the parts in the LBM are standard, such as f^{eq}, streaming, calculation, macroscopic variables calculation, successfully conducting LBM simulation is highly depended on boundary condition implementation.

BC is important for accuracy, stability and efficiency of the LBM.

1.4 Boundary condition of the LBM



- Heuristic BC: periodic BC, symmetrical BC, full developed BC, bounce-back BC, specular reflection BC
- Kinetic BC: Zou-He BC, counter-slip BC
- Extrapolation scheme
- BC for curved boundary

1.4 Boundary condition of the LBM





Bottom boundary



Left boundary



Right boundary

1.4 Boundary condition of the LBM

Bounce back **Non-slip boundary**



Standard bounce back	$f_{2,5,6}(i,j,t)=f_{4,7,8}(i,j,t)$	**1 order**
Modified bounce back	$f'_{2,5,6}(i,j,t)=f_{4,7,8}(i,j,t)$	**2 order**
Half-way bounce back	$f_{2,5,6}(i,j,t)=f_{4,7,8}(i,j,t)$	**2 order**

1.4 Boundary condition of the LBM



$$f_{1,5,8}(1, j, t) = f_{1,5,8}(\text{nx}, j, t)$$

$$f_{3,6,7}(\text{nx}, j, t) = f_{3,6,7}(1, j, t)$$

1.4 Boundary condition of the LBM

Zou-He boundary condition **Velocity or pressure is known**



$$f_1 + f_5 + f_8 - f_3 - f_6 - f_7 = u_x$$

$$f_2 + f_5 + f_6 - f_4 - f_7 - f_8 = u_y$$

$$f_1 - f_1^{eq} = f_3 - f_3^{eq}$$

1.4 Boundary condition of the LBM

Extrapolation



$$f_2(\mathrm{i},1) = f_2^{\mathrm{eq}}(\mathrm{i},1)\left(\rho(i,2),\mathrm{u}(\mathrm{i},1)\right) + f_2(i,2) - f_2^{\mathrm{eq}}(i,2)$$

$$f_2(\mathrm{i},1) = f_2^{\mathrm{eq}}(\mathrm{i},1)\left(\rho(i,1),\mathrm{u}(\mathrm{i},2)\right) + f_2(i,2) - f_2^{\mathrm{eq}}(i,2)$$

1.4 Boundary condition of the LBM

Curved boundary

Interpolation/extrapolation method

$$f_i^*(\mathbf{x}_s) = (1-\chi) f_i^*(\mathbf{x}_f) + \chi f_i^*(\mathbf{x}_b)$$

$$\triangle = |\mathbf{x}_f - \mathbf{x}_b| / |\mathbf{x}_f - \mathbf{x}_s|$$

$$f_i^*(\mathbf{x}_b) = w_i \rho_f \left[1 + \frac{\mathbf{u}_{bf}\cdot\mathbf{c}_i}{c_s^2} + \frac{(\mathbf{u}_f\cdot\mathbf{c}_i)^2}{2c_s^4} - \frac{u_f^2}{2c_s^2} \right]$$



Filippova and D. Hanel, JCP, 147, 219 (1998）

$$\mathbf{u}_{bf} = (\triangle - 1)\mathbf{u}_f/\triangle + \mathbf{u}_b/\triangle \qquad \chi = (2\triangle - 1)/\tau \qquad \triangle \geq 1/2$$

$$\mathbf{u}_{bf} = \mathbf{u}_f, \quad \chi = (2\triangle - 1)/(\tau - 1) \qquad \chi = (2\triangle - 1)/(\tau - 1) \qquad \triangle < 1/2.$$

Content

1.5 External force implementation

He-Shan-Doolen model

$$\frac{\partial f}{\partial t}+\boldsymbol{\xi}\cdot\frac{\partial f}{\partial \mathbf{x}}+\mathbf{a}\cdot\frac{\partial f}{\partial \boldsymbol{\xi}}=-\frac{1}{\tau}(f-f^{\mathrm{eq}})$$

$$\mathbf{a}\cdot\frac{\partial f}{\partial \boldsymbol{\xi}}\approx\mathbf{a}\cdot\frac{\partial f^{\mathrm{eq}}}{\partial \boldsymbol{\xi}}$$

$$=\mathbf{a}\cdot\frac{\partial}{\partial \boldsymbol{\xi}}\left(\frac{1}{(2\pi RT)^{3/2}}\exp(-\frac{(\boldsymbol{\xi}-\mathbf{u})^2}{2RT})\right)$$

$$=-\frac{(\boldsymbol{\xi}-\mathbf{u})\cdot\mathbf{a}}{RT}f^{\mathrm{eq}}$$

$$\rho\mathbf{u}=\sum_i \mathbf{c}_i f_i+\frac{1}{2}\mathbf{F}\delta t$$

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t,t+\Delta t)-f_i(\mathbf{x},t)=-\frac{1}{\tau}(f_i(\mathbf{x},t)-f_i^{\mathrm{eq}}(\mathbf{x},t))+(1-\frac{1}{2\tau})\frac{(\mathbf{c}_i-\mathbf{u})\cdot\mathbf{a}}{c_s^{\,2}}f^{\mathrm{eq}}$$

Content

1.6 LBM program structure



$$f \rightarrow f^{eq}$$

$$f' \rightarrow f - \frac{1}{\tau}(f - f^{eq})$$

$$f_i(\mathbf{x} + \mathbf{c}_i \Delta t, t + \Delta t) = f'$$

$$\rho = \sum_{i=0} f_i, \quad \rho\mathbf{u} = \sum_{i=0} f_i \mathbf{e}_i .$$

1.6 LBM program structure

The LB model adopted is the imcompressible model developed by Prof. Zhaoli Guo in 2000.

Lattice BGK Model for Incompressible Navier–Stokes Equation, Journal of Computational Physics, 165(1), 2000, Pages 288-306.

The code can be downloaded at: http://nht.xjtu.edu.cn/down.asp



首页 | 小组简介 | 新闻公告 | 研究人员 | 研究成果 | 科研动态 | 资料下载 | 小组活动 | 人才需求

资料下载

当前位置：首页>>资料下载

	资料名称	发布时间	下载次数
1.	2018《计算传热学的近代进展》第五章	2018/5/16	51
2.	2018《计算传热学的近代进展》第四章	2018/5/15	39
3.	2018《计算传热学的近代进展》第三章	2018/5/11	54
4.	LBM流动程序	2018/5/7	134
5.	LBM传质程序	2018/5/6	94

```
!==========================================================================
!    This code was written by Li Chen at Xi'an Jiaotong University.
!    This code was for single-phase in a 2D channel.
!    Pressures are known at the left inlet and right outlet,
!    and non_slip conditions at the top and bottom walls.
!    One can refer to the following papers for more details:
!    Li Chen et al., Water Resources Research Volume: 50(12): 9343-9365, 2014
!    Li Chen: lichennht08@mail.xjtu.edu.cn.
!==========================================================================
```

```
!==========================================
MODULE START_L
  PARAMETER (nx=21,ny=81)
  integer::I,J,K,LAST,ITER
  double precision,PARAMETER::XL=20.E-5,YL=80.E-5
  double precision,dimension(nx)::X
  double precision,dimension(ny)::Y
  double precision,dimension(0:nx+1,0:ny+1)::U,V,PRE
!----------------------------------------
  double precision::DX,DY,DT
  double precision,PARAMETER::C=1.d0,CS2=1.d0/3.d0
  integer,dimension(0:8)::FCX=(/0,1,0,-1,0,1,-1,-1,1/)
  integer,dimension(0:8)::FCY=(/0,0,1,0,-1,1,1,-1,-1/)
  double precision,dimension(0:8)::wi
  double precision,dimension(0:2)::lambda
  !--------------------------------------------------
  double precision,dimension(0:8,0:nx+1,0:ny+1)::f1,ff1
  double precision::ftao,vmu_phy,vmu_lat,preleft,preright,feq1
!----------------------------------------------------
  integer,dimension(0:nx+1,0:ny+1)::ls
  logical,dimension(0:nx+1,0:ny+1)::walls
  double precision::delta
  double precision::sumc_last,sumu
 END MODULE
!==========================================
```

Velocity and pressure

Lattice velocity and sound speed

DnQm model

Parameters in LB model

PDF

Define of solid structure

```
!===========================================================
PROGRAM MAIN
USE START_L

    CALL SOLID_STRUCTURE
    CALL INITIALIZATION

    DO iter=1,last
        CALL COLLISIONF
        CALL STREAMF
        CALL BOUNDARYF
        CALL MACROF
        if(mod(iter,1000).eq.0) CALL OUTPUT
    ENDDO

END PROGRAM
!============================================================
```

```
!=================================================================-=
SUBROUTINE SOLID_STRUCTURE
USE START_L
    ! ls represents the porous structure: 0 denotes nodes of void space, 1 denotes solid node.
    ls=0
    ls(:,ny:ny+1)=1
    ls(:,0:1)=1
    walls=.false.
    do j=0,ny+1
    do i=0,nx+1
        if(ls(i,j).eq.1) then
            walls(i,j)=.true.
        endif
    enddo
    enddo
RETURN
END SUBROUTINE
!==================================================================
```

Here, you can input the solid structures you want to simulate!
Input the structure date. A 2D matrix with 0 for fluid and 1 for solid.

See Slides for porous flow!

```fortran
!===========================================================
SUBROUTINE INITIALIZATION
USE START_L
double precision::z1,z2

    dx=xl/float(nx-1)
    dy=dx
    last=500000
    delta=1.d0
    lambda(0)=-5.d0/3.d0
    lambda(1)=1.0d0/3.d0
    lambda(2)=1.d0/12.d0
    wi(0)=4.d0/9.d0
    wi(1:4)=1.d0/9.d0
    wi(5:8)=1.d0/36.d0
    vmu_phy=20.e-6
    ftao=1.d0
    vmu_lat=(ftao-0.5d0)/3.d0
    scale=vmu_phy/vmu_lat
    dt=dx**2./scale
    preleft=1.0002d0
    preright=1.d0

    do j=1,ny
    do i=1,nx
       pre(i,j)=preleft-float(i-1)/float(nx-1)*(preleft-preright)
       u(i,j)=0.d0
       v(i,j)=0.d0
    enddo
    enddo
```

Physical length of a lattice

Parameters in LB model.

Viscosity in physical units

Relaxation time

Viscosity in lattice units

Physical of one lattice iteration step.

Pressure difference between inlet and outlet.

Velocity and pressure initialization

$$f_i^{eq} = \begin{cases} -4\sigma \dfrac{p}{c^2} + s_i(\mathbf{u}) & (i=0) \\ \lambda \dfrac{p}{c^2} + s_i(\mathbf{u}) & (i=1-4) \\ \gamma \dfrac{p}{c^2} + s_i(\mathbf{u}) & (i=5-8) \end{cases}$$

σ=5/12
λ=1/3
γ=1/12

```fortran
  do j=1,ny
  do i=1,nx
     z2=u(i,j)**2.d0+v(i,j)**2.d0
     do k=0,8
        z1=fcx(k)*u(i,j)+fcy(k)*v(i,j)
        if(k.eq.0) then
           feq1=lambda(0)*pre(i,j)+wi(k)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
        elseif(k.le.4.and.k.ge.1) then
           feq1=lambda(1)*pre(i,j)+wi(k)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
        elseif(k.le.8.and.k.ge.5) then
           feq1=lambda(2)*pre(i,j)+wi(k)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
        endif
        f1(k,i,j)=feq1
        ff1(k,i,j)=feq1
     enddo
  enddo
  enddo

RETURN
END SUBROUTINE
!=========================================================
```

$$f_i^{eq}=\begin{cases} -4\sigma\dfrac{p}{c^2}+s_i(\mathbf{u}) & (i=0) \\ \lambda\dfrac{p}{c^2}+s_i(\mathbf{u}) & (i=1-4) \\ \gamma\dfrac{p}{c^2}+s_i(\mathbf{u}) & (i=5-8) \end{cases}$$

$$s_i(\mathbf{u})=\omega_i\left[3\frac{\mathbf{c}_i\cdot\mathbf{u}}{c}+4.5\frac{(\mathbf{c}_i\cdot\mathbf{u})^2}{c^2}-1.5\frac{\mathbf{u}\cdot\mathbf{u}}{c^2}\right]$$

Distribution function initialization!

```fortran
!=======================================================================
SUBROUTINE COLLISIONF
USE START_L
double precision::z1,z2

   do j=1,ny
   do i=1,nx
      if(.not.walls(i,j)) then
      z2=u(i,j)**2.+v(i,j)**2.
      do k=0,8
         z1=fcx(k)*u(i,j)+fcy(k)*v(i,j)
         if(k.eq.0) then
            feq1=lambda(0)*pre(i,j)+wi(k)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
         elseif(k.le.4.and.k.ge.1) then
            feq1=lambda(1)*pre(i,j)+wi(k)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
         elseif(k.le.8.and.k.ge.5) then
            feq1=lambda(2)*pre(i,j)+wi(k)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
         endif
         ff1(k,i,j)=f1(k,i,j)-1.d0/ftao*(f1(k,i,j)-feq1)
     enddo
    endif
   enddo
   enddo

RETURN
END SUBROUTINE
!=======================================================================
```

$$f' \rightarrow f - \frac{1}{\tau}(f - f^{eq})$$

```
!=========================================
SUBROUTINE STREAMF
USE START_L

!-------------------periodic boundary along y---------------
    do j=1,ny
    do i=1,nx
    do k=0,8
        f1(k,i,j)=ff1(k,i-int(fcx(k)),j-int(fcy(k)))
    enddo
    enddo
    enddo

RETURN
END SUBROUTINE
!=========================================
```

$$f_i(\mathbf{x}+\mathbf{c}_i\Delta t, t+\Delta t) = f\,'$$

```
!=========================================================================
SUBROUTINE BOUNDARYF
USE START_L

  do j=1,ny
  do i=1,nx
    if(walls(i,j)) then
     ff1(1,i,j)=f1(3,i,j)
     ff1(3,i,j)=f1(1,i,j)
     ff1(2,i,j)=f1(4,i,j)
     ff1(4,i,j)=f1(2,i,j)
     ff1(5,i,j)=f1(7,i,j)
     ff1(7,i,j)=f1(5,i,j)
     ff1(6,i,j)=f1(8,i,j)
     ff1(8,i,j)=f1(6,i,j)
    endif
  enddo
  enddo

  do j=1,ny
    z1=fcx(1)*u(2,j)+fcy(1)*v(2,j)
    z2=u(2,j)**2.d0+v(2,j)**2.d0
    feq1=lambda(1)*pre(2,j)+wi(1)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
    f1(1,1,j)=lambda(1)*pre(1,j)+wi(1)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2) +F1(1,2,J)-FEQ1
    z1=fcx(5)*u(2,j)+fcy(5)*v(2,j)
    feq1=lambda(2)*pre(2,j)+wi(5)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
    f1(5,1,j)=lambda(2)*pre(1,j)+wi(5)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)+F1(5,2,J)-FEQ1
    z1=fcx(8)*u(2,j)+fcy(8)*v(2,j)
    feq1=lambda(2)*pre(2,j)+wi(8)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
    f1(8,1,J)=lambda(2)*pre(1,j)+wi(8)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)+F1(8,2,J)-FEQ1
 enddo
```

非平衡外推！

Non-equilibrium extrapolation method for velocity and pressure boundary conditions in the lattice Boltzmann method
ZL Guo, CG Zheng, BC Shi
Chinese Physics 11 (4), 366

```
do j=1,ny
     z1=fcx(3)*u(nx-1,J)+fcy(3)*v(nx-1,j)
     z2=u(nx-1,j)**2.d0+v(nx-1,j)**2.d0
     feq1=lambda(1)*pre(nx-1,j)+wi(3)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
     f1(3,nx,j)=lambda(1)*pre(nx,j)+wi(3)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2) +F1(3,nx-1,J)-FEQ1
     z1=fcx(6)*u(nx-1,j)+fcy(6)*v(nx-1,j)
     feq1=lambda(2)*pre(nx-1,j)+wi(6)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
     f1(6,nx,j)=lambda(2)*pre(nx,j)+wi(6)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2) +F1(6,nx-1,J)-FEQ1
     z1=fcx(7)*u(nx-1,j)+fcy(7)*v(nx-1,j)
     feq1=lambda(2)*pre(nx-1,j)+wi(7)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2)
     f1(7,nx,j)=lambda(2)*pre(nx,j)+wi(7)*(3.d0*z1+4.5d0*z1**2.-1.5d0*z2) +F1(7,nx-1,J)-FEQ1
 enddo

RETURN
END SUBROUTINE
!===============================================================
```

```
!==========================================
SUBROUTINE MACROF
USE START_L

   do j=1,ny
   do i=1,nx
      IF(.not.walls(i,j)) THEN
         temppre=0.0d0
         tempu=0.0d0
         tempv=0.0d0
         do k=1,8
            temppre=temppre+f1(k,i,j)
            tempu=tempu+f1(k,i,j)*fcx(k)
            tempv=tempv+f1(k,i,j)*fcy(k)
         enddo
         u(i,j)=tempu
         v(i,j)=tempv
         temp1=u(i,j)**2.d0+v(i,j)**2.d0
         pre(i,j)=(temppre-2.d0/3.d0*temp1)/(-lambda(0))
      elseif(walls(i,j))then
         u(i,j)=0.d0
         v(i,j)=0.d0
         pre(i,j)=0.d0
      endif
   enddo
   enddo

   do j=1,ny
      pre(1,j)=preleft
      pre(nx,j)=preright
   enddo

RETURN
END SUBROUTINE
```

$$\mathbf{u}=\sum_{i=1}^{8}\mathbf{c}_i f_i,\quad \frac{p}{\rho}=\frac{c^2}{4\sigma}\left[\sum_{i=1}^{8} f_i+s_0(u)\right]$$

$$s_i(\mathbf{u})=\omega_i\left[3\frac{\mathbf{c}_i\cdot\mathbf{u}}{c}+4.5\frac{(\mathbf{c}_i\cdot\mathbf{u})^2}{c^2}-1.5\frac{\mathbf{u}\cdot\mathbf{u}}{c^2}\right]$$

$S_0(\mathbf{u})$

```
!========================================================================
SUBROUTINE OUTPUT
USE START_L

    sumu_last=sumu
    sumu=0.d0
    do j=1,ny
    do i=1,nx
        if(.not.walls(i,j)) sumu=sumu+u(i,j)
    enddo
    enddo
    delta=abs(sumu_last-sumu)/abs(sumu)
    write(*,*) iter,u(nx-10,ny/2),delta

    open(10,file="velocity_pressure.dat")
    write(10,*)'VARIABLES= X,Y,u,v,pre'
    WRITE(10,*)'ZONE I=',nx,',J=',ny,',T=TT'
    do j=1,ny
    do i=1,nx
        write(10,*) i,j,u(i,j),v(i,j),pre(i,j)
    enddo
    enddo
    close(10)

    if(delta.le.1.e-8) stop

RETURN
END SUBROUTINE
!========================================================================
```

华人的贡献

❖在LBM二十余年的发展历史中，华人科学家做出了重要的贡献。LBM被他引次数最高的文献中华人的参与非常高。

Lattice Boltzmann method for fluid flows
S Chen, GD Doolen - Annual review of fluid mechanics, 1998 - annualreviews.org
▪ Abstract We present an overview of the lattice Boltzmann method (LBM), a parallel and efficient algorithm for simulating single-phase and multiphase fluid flows and for incorporating additional physical complexities. The LBM is especially useful for modeling ...
☆ 99 被引用次数：6380 相关文章 所有 13 个版本

Theory of the lattice Boltzmann method: From the Boltzmann equation to the lattice Boltzmann equation
X He, LS Luo - Physical Review E, 1997 - APS
In this paper, the lattice Boltzmann equation is directly derived from the Boltzmann equation. It is shown that the lattice Boltzmann equation is a special discretized form of the Boltzmann equation. Various approximations for the discretization of the Boltzmann equation in both ...
☆ 99 被引用次数：1451 相关文章 所有 11 个版本

Discrete lattice effects on the forcing term in the lattice Boltzmann method
Z Guo, C Zheng, B Shi - Physical Review E, 2002 - APS
We show that discrete lattice effects must be considered in the introduction of a force into the lattice Boltzmann equation. A representation of the forcing term is then proposed. With the representation, the Navier-Stokes equation is derived from the lattice Boltzmann equation ...
☆ 99 被引用次数：1153 相关文章 所有 11 个版本

A novel thermal model for the lattice Boltzmann method in incompressible limit
X He, S Chen, GD Doolen - Journal of Computational Physics, 1998 - Elsevier
A novel lattice Boltzmann thermal model is proposed for studying thermohydrodynamics in incompressible limit. The new model introduces an internal energy density distribution function to simulate the temperature field. The macroscopic density and velocity fields are ...
☆ 99 被引用次数：1221 相关文章 所有 10 个版本

Theory of the lattice Boltzmann method: Dispersion, dissipation, isotropy, Galilean invariance, and stability

P Lallemand, LS Luo - Physical Review E, 2000 - APS

The generalized hydrodynamics (the wave vector dependence of the transport coefficients) of a generalized lattice Boltzmann equation (LBE) is studied in detail. The generalized lattice Boltzmann equation is constructed in moment space rather than in discrete velocity space ...

☆ 99 被引用次数：1398 相关文章 所有 15 个版本

Simulation of multicomponent fluids in complex three-dimensional geometries by the lattice Boltzmann method

NS Martys, H Chen - Physical review E, 1996 - APS

We describe an implementation of the recently proposed lattice Boltzmann based model of Shan and Chen [Phys. Rev. E 47, 1815 (1993); 49, 2941 (1994)] to simulate multicomponent flow in complex three-dimensional geometries such as porous media. The above method ...

☆ 99 被引用次数：806 相关文章 所有 8 个版本

Lattice-Boltzmann method for complex flows

CK Aidun, JR Clausen - Annual review of fluid mechanics, 2010 - annualreviews.org

With its roots in kinetic theory and the cellular automaton concept, the lattice-Boltzmann (LB) equation can be used to obtain continuum flow quantities from simple and local update rules based on particle interactions. The simplicity of formulation and its versatility explain the ...

☆ 99 被引用次数：1262 相关文章 所有 6 个版本

Simulation of cavity flow by the lattice Boltzmann method

S Hou, Q Zou, S Chen, GD Doolen... - arXiv preprint comp-gas ..., 1994 - arxiv.org

A detailed analysis is presented to demonstrate the capabilities of the lattice Boltzmann method. Thorough comparisons with other numerical solutions for the two-dimensional, driven cavity flow show that the lattice Boltzmann method gives accurate results over a wide ...

☆ 99 被引用次数：731 相关文章 所有 12 个版本 »

An extrapolation method for boundary conditions in lattice Boltzmann method

Z Guo, C Zheng, B Shi - Physics of Fluids, 2002 - aip.scitation.org

A boundary treatment for curved walls in lattice Boltzmann method is proposed. The distribution function at a wall node who has a link across the physical boundary is decomposed into its equilibrium and nonequilibrium parts. The equilibrium part is then ...

☆ 99 被引用次数：617 相关文章 所有 4 个版本

Lattice Boltzmann model for simulating flows with multiple phases and components
X Shan, H Chen - Physical Review E, 1993 - APS
A lattice Boltzmann model is developed which has the ability to simulate flows containing multiple phases and components. Each of the components can be immiscible with the others and can have different mass values. The equilibrium state of each component can have a ...
☆ 99 被引用次数：2452 相关文章 所有 11 个版本

Simulation of Rayleigh-Bénard convection using a lattice Boltzmann method
X Shan - Physical Review E, 1997 - APS
Rayleigh-Bénard convection is numerically simulated in two and three dimensions using a recently developed two-component lattice Boltzmann equation (LBE) method. The density field of the second component, which evolves according to the advection-diffusion equation ...
☆ 99 被引用次数：532 相关文章 所有 13 个版本

Simulation of nonideal gases and liquid-gas phase transitions by the lattice Boltzmann equation
X Shan, H Chen - Physical Review E, 1994 - APS
We describe in detail a recently proposed lattice-Boltzmann model [X. Shan and H. Chen, Phys. Rev. E 47, 1815 (1993)] for simulating flows with multiple phases and components. In particular, the focus is on the modeling of one-component fluid systems which obey nonideal ...
☆ 99 被引用次数：1003 相关文章 所有 17 个版本

Multicomponent lattice-Boltzmann model with interparticle interaction
X Shan, G Doolen - Journal of Statistical Physics, 1995 - Springer
A lattice Boltzmann model for simulating fluids with multiple components and interparticle forces proposed by Shan and Chen is described in detail. Macroscopic equations governing the motion of each component are derived by using the Chapman-Enskog method. The ...
☆ 99 被引用次数：484 相关文章 所有 10 个版本

Equations of state in a lattice Boltzmann model
P Yuan, L Schaefer - Physics of Fluids, 2006 - aip.scitation.org
In this paper we consider the incorporation of various equations of state into the single-component multiphase lattice Boltzmann model. Several cubic equations of state, including the van der Waals, Redlich-Kwong, and Peng-Robinson, as well as a noncubic equation of ...
☆ 99 被引用次数：401 相关文章 所有 5 个版本

Work of NHT

Lattice Boltzmann method for gaseous microflows using kinetic theory boundary conditions GH Tang, WQ Tao, YL He Physics of Fluids 17 (5), 058101	143	2005
Thermal boundary condition for the thermal lattice Boltzmann equation GH Tang, WQ Tao, YL He Physical Review E 72 (1), 016703	131	2005
Electroosmotic flow of non-Newtonian fluid in microchannels GH Tang, XF Li, YL He, WQ Tao Journal of Non-Newtonian Fluid Mechanics 157 (1-2), 133-137	130	2009
Pore-scale flow and mass transport in gas diffusion layer of proton exchange membrane fuel cell with interdigitated flow fields L Chen, HB Luan, YL He, WQ Tao International Journal of Thermal Sciences 51, 132-144	116	2012
Nanoscale simulation of shale transport properties using the lattice Boltzmann method: permeability and diffusivity L Chen, L Zhang, Q Kang, J Yao, W Tao Scientific Report 5 (8089)	114	2015
Lattice Boltzmann modeling of microchannel flows in the transition flow regime Q Li, YL He, GH Tang, WQ Tao Microfluidics and nanofluidics 10 (3), 607-618	91	2011
Pore-scale modeling of multiphase reactive transport with phase transitions and dissolution-precipitation processes in closed systems L Chen, Q Kang, BA Robinson, YL He, WQ Tao Physical Review E 87 (4), 043306	80	2013
A critical review of the pseudopotential multiphase lattice Boltzmann model: Methods and applications L Chen, Q Kang, Y Mu, YL He, WQ Tao International Journal of Heat and Mass Transfer 76, 210-236	202	2014

Review

Reference

Chen, S. Y., and G. D. Doolen (1998), Lattice Boltzmann methode for fluid flows, Annual Review of Fluid Mechanics, 30, 329-364.

Succi, S. (2001), The lattice Boltzmann equation: for fluid dynamics and beyond, Oxford University Press, OXford.

Sukop, M. C., and D. T. J. Thorne (2006), Lattice Boltzmann Modeling: An Introduction for Geoscientists and Engineers, Springer Publishing Company, New York.

Zhang, J. (2011), Lattice Boltzmann method for microfluidics: models and applications, Microfluid Nanofluid (10, 1-28).

郭照立，郑楚光，格子**Boltzmann**方法的原理及应用，科学出版社，**2009**

何雅玲，王勇，李庆，格子**Boltzmann**方法的理论及应用，科学出版社，**2009**